



CS334 计算机组成实验

LAB6

简单的类 MIPS 多周期流水化处理器

周鼎
5140219268

指导老师：王老师

April 7, 2016

Contents

I	基本的流水线实现	3
1	五级流水线的顶层结构	3
2	系统定义顶层所需的寄存器	3
3	分阶段定义网线类型并完成实例化和连接	4
3.1	IF Stage	4
3.2	ID Stage	5
3.3	EX & MEM	7
3.4	阶段间的寄存器更新	8
4	ModelSim 仿真	9
II	改进	10
5	Flush	10
5.1	改进方案	10
5.2	ModelSim 仿真	11
6	JUMP	11
6.1	改进方案	11
6.2	ModelSim 仿真	12
7	数据冒险	13
7.1	MUX	13
7.2	Forward Unit	14
7.3	连接信号线及实例化	15
7.4	ModelSim 仿真	15
III	上板试验	16
8	IO Scheme	16
9	IO 控制逻辑	17
10	User Constraint File	17

Part I

基本的流水线实现

1 五级流水线的顶层结构

将指令执行分为五个阶段，每个阶段每条指令只使用一个主要模块，如下图所示，相比单周期，每阶段间添加寄存器存储必要的信息。




Figure 1: 顶层设计

2 系统定义顶层所需的寄存器

流水线的顶层中需要较多的寄存器，为了做到不重不漏，按阶段和用途分类定义。分为 IF/ID, ID/EX, EX/MEM, MEM/WB 四个大块，内分为控制信号和指令信息，指令信息里有 PC+4，也有后续指令执行过程得到的结果；控制信号按信号使用的阶段分类，具体见如下代码。

```

1 //1.0 For stage IF to ID;
2 reg [31:0] IF_ID_PCAAdd4;
3 reg [31:0] IF_ID_Instr;
4
5 //2.0 For stage ID to EX;
6 reg [31:0] ID_EX_PCAAdd4;
7 reg [31:0] ID_EX_RegReadData1;
8 reg [31:0] ID_EX_RegReadData2;
9 reg [31:0] ID_EX_SignExt;
10 reg [20:16] ID_EX_InstHigh;
11 reg [15:11] ID_EX_InstLow;
12 //2.1 To EX

```

```

13 reg ID_EX_RegDst;
14 reg [1:0] ID_EX_ALUOp;
15 reg ID_EX_ALUSrc;
16 //2.2 To MEM
17 reg ID_EX_Branch;
18 reg ID_EX_MemWrite;
19 reg ID_EX_MemRead;
20 //2.3 To WB
21 reg ID_EX_MemToReg;
22 reg ID_EX_RegWrite;
23
24 //3.0 For stage EX to MEM;
25 reg EX_MEM_Zero;
26 reg [31:0] EX_MEM_ALUOut;
27 reg [31:0] EX_MEM_BranchAddress;
28 reg [31:0] EX_MEM_RegReadData2;
29 //3.1 To MEM
30 reg EX_MEM_Branch;
31 reg EX_MEM_MemWrite;
32 reg EX_MEM_MemRead;
33 reg [4:0] EX_MEM_RegWriteAddress; //rt or rd
34 //3.2 To WB
35 reg EX_MEM_MemToReg;
36 reg EX_MEM_RegWrite;
37
38 //4.0 For stage MEM to WB;
39 reg [31:0] MEM_WB_ALUOut;
40 reg [31:0] MEM_WB_MemReadData;
41 reg [4:0] MEM_WB_RegWriteAddress;
42 reg MEM_WB_RegWrite;
43 reg MEM_WB_MemToReg;

```

reg_def_top.v

3 分阶段定义网线类型并完成实例化和连接

3.1 IF Stage

IF 阶段的实现要特别注意线型变量不要和 WB 阶段混淆，以及 NEXT_PC 的实现。

```

1 //1.0 IF
2 wire IF_PCSrc, //for MUX sel signal
3 IF_Branch,
4 IF_Zero;
5 wire [31:0] IF_BranchAddress;
6 wire [31:0] IF_CurrPC;
7 wire [31:0] IF_NextPC;
8 wire [31:0] IF_PCAdd4;
9 wire [31:0] IF_Instr;
10 //associate with reg
11 assign IF_BranchAddress = EX_MEM_BranchAddress;
12 assign IF_Zero = EX_MEM_Zero;
13 assign IF_Branch = EX_MEM_Branch;
14 //Combinational Logic
15 assign IF_PCAdd4 = IF_CurrPC +4;
16 assign IF_PCSrc = IF_Branch & IF_Zero & ~RESET;
17 assign IF_NextPC = IF_PCSrc? IF_BranchAddress: IF_PCAdd4; //MUX
18 //assign IF_NextPC = IF_PCAdd4;
19 //update PC at posedge

```

```

20 Pc mainPC (
21   .clock_in(CLK),
22   .nextPC(IF_NextPC),
23   .currPC(IF_CurrPC),
24   .rst(RESET)
25 );
26 wire [31:0] IF_Index;
27 assign IF_Index = IF_CurrPC>>2;
28 instructionMemory InstrMemory (
29   .address(IF_Index),
30   .clock_in(CLK),
31   .reset(RESET),
32   .readData(IF_Instr)
33 );
34 //InstMem;
35

```

IF.v

3.2 ID Stage

ID 阶段所需要的主控信号与单周期的功能完全一样，可直接使用，唯一不同的是产生的控制信号需要用寄存器存储至相应阶段使用。




Figure 2: 后三阶段的控制信号

```

1 wire [5:0] ID_OpCode;
2 wire [31:0] ID_PCAdd4;
3 wire [4:0] ID_RegReadAddress1;
4 wire [4:0] ID_RegReadAddress2;
5 wire [15:0] ID_Imm;
6 wire [31:0] ID_SignExt;
7 wire [20:16] ID_InstHigh;
8 wire [15:11] ID_InstLow;
9 wire [31:0] ID_RegReadData1;
10 wire [31:0] ID_RegReadData2;
11 wire WB_RegWrite;
12 wire WB_MemToReg;
13 wire [4:0] WB_RegWriteAddress;
14 wire [31:0] WB_MemReadData;

```

```

15 wire [31:0] WB_ALUOut;
16 wire [31:0] WB_RegWriteData;
17 //associate with reg
18 assign ID_Instr = IF_ID_Instr;
19 assign ID_PCAdd4 = IF_ID_PCAdd4;
20 assign ID_OpCode = ID_Instr[31:26];
21 assign ID_RegReadAddress1 = ID_Instr[25:21];
22 assign ID_RegReadAddress2 = ID_Instr[20:16];
23 assign ID_Imm = ID_Instr[15:0];
24 assign ID_InstHigh = ID_Instr[20:16];
25 assign ID_InstLow = ID_Instr[15:11];
26 assign WB_RegWrite = MEM_WB_RegWrite;
27 assign WB_MemToReg = MEM_WB_MemToReg;
28 assign WB_RegWriteAddress = MEM_WB_RegWriteAddress;
29 assign WB_ALUOut = MEM_WB_ALUOut;
30 assign WB_MemReadData = MEM_WB_MemReadData;
31 assign WB_RegWriteData = WB_MemToReg? WB_MemReadData: WB_ALUOut;
32
33 Register mainReg (
34     .clock_in(CLK),
35     .regWrite(WB_RegWrite),
36     .readReg1(ID_RegReadAddress1),
37     .readReg2(ID_RegReadAddress2),
38     .writeReg(WB_RegWriteAddress),
39     .writeData(WB_RegWriteData),
40     .reset(RESET),
41     .readData1(ID_RegReadData1),
42     .readData2(ID_RegReadData2),
43     .ioinput(SW),
44     .iooutput(LED)
45 );
46
47 //2.1 To EX
48 wire [1:0] ID_ALUOp;
49 wire ID_RegDst;
50 wire ID_ALUSrc;
51 //2.2 To MEM
52 wire ID_Branch;
53 wire ID_MemWrite;
54 wire ID_MemRead;
55 //2.3 To WB
56 wire ID_MemToReg;
57 wire ID_RegWrite;
58 wire JUMP; //of no use
59 Ctr mainCtr (
60     .opCode(ID_OpCode),
61     .regDst(ID_RegDst),
62     .aluSrc(ID_ALUSrc),
63     .memToReg(ID_MemToReg),
64     .regWrite(ID_RegWrite),
65     .memRead(ID_MemRead),
66     .memWrite(ID_MemWrite),
67     .branch(ID_Branch),
68     .aluOp(ID_ALUOp),
69     .jump(JUMP)
70 );
71 signExt mainSignExt (
72     .inst(ID_Imm),
73     .data(ID_SignExt)
74 );

```

ID.v

3.3 EX & MEM

后面的过程与单周期十分类似，不在赘述，仅列出代码。

```

1  //2.0 For stage ID to EX;
2  wire [31:0] EX_PCAAdd4;
3  wire [31:0] EX_ALUSrc1;
4  wire [31:0] EX_ALUSrc2;
5  wire [31:0] EX_RegReadData2;
6  wire [31:0] EX_SignExt;
7  wire [20:16] EX_InstHigh;
8  wire [15:11] EX_InstLow;
9  wire EX_RegDst,
10     EX_ALUSrc;
11  wire [1:0] EX_ALUOp;
12  wire [4:0] EX_RegWriteAddress;
13  wire [5:0] EX_Funct;
14  //associate with Reg
15  assign EX_RegDst = ID_EX_RegDst;
16  assign EX_ALUOp = ID_EX_ALUOp;
17  assign EX_ALUSrc1 = ID_EX_RegReadData1;
18  assign EX_ALUSrc = ID_EX_ALUSrc; //control signal
19  assign EX_ALUSrc2 = EX_ALUSrc? EX_SignExt: EX_RegReadData2; //MUX
20  assign EX_RegWriteAddress = EX_RegDst? EX_InstLow: EX_InstHigh;
21  assign EX_PCAAdd4 = ID_EX_PCAAdd4;
22  assign EX_RegReadData2 = ID_EX_RegReadData2;
23  assign EX_SignExt = ID_EX_SignExt;
24  assign EX_Funct = ID_EX_SignExt[5:0];
25  assign EX_InstHigh = ID_EX_InstHigh;
26  assign EX_InstLow = ID_EX_InstLow;
27  //Instances
28  wire [3:0] EX_ALUCtr;
29  AluCtr mainALUCtr (
30      .aluOp(EX_ALUOp),
31      .funct(EX_Funct),
32      .aluCtr(EX_ALUCtr)
33  );
34  wire EX_Zero;
35  wire [31:0] EX_ALUOut;
36  Alu mainALU (
37      .input1(EX_ALUSrc1),
38      .input2(EX_ALUSrc2),
39      .aluCtr(EX_ALUCtr),
40      .zero(EX_Zero),
41      .aluRes(EX_ALUOut)
42  );
43  //2.2 To MEM
44  wire EX_Branch,
45      EX_MemWrite,
46      EX_MemRead;
47  assign EX_Branch = ID_EX_Branch;
48  assign EX_MemWrite = ID_EX_MemWrite;
49  assign EX_MemRead = ID_EX_MemRead;
50  //2.3 To WB
51  wire EX_MemToReg,
52      EX_RegWrite;
53  assign EX_MemToReg = ID_EX_MemToReg;
54  assign EX_RegWrite = ID_EX_RegWrite;
55  //BranchAddress
56  wire [31:0] EX_BranchAddress;
57  assign EX_BranchAddress = (EX_SignExt<<2) + EX_PCAAdd4;

```

EX.v

```

1 //4.0 MEM
2 wire MEM_MemWrite,
3   MEM_MemRead;
4 wire [4:0] MEM_RegWriteAddress;
5 wire [31:0] MEM_ALUOut;
6 wire [31:0] MEM_MemWriteData;
7 wire [31:0] MEM_MemReadData;
8 //associate with reg
9 assign MEM_MemWrite = EX_MEM_MemWrite;
10 assign MEM_MemRead = EX_MEM_MemRead;
11 assign MEM_RegWriteAddress = EX_MEM_RegWriteAddress;
12 assign MEM_ALUOut = EX_MEM_ALUOut;
13 assign MEM_MemWriteData = EX_MEM_RegReadData2;
14 //instances
15 dataMemory DataMemory (
16   .clock_in(CLK),
17   .address(MEM_ALUOut),
18   .writeData(MEM_MemWriteData),
19   .readData(MEM_MemReadData),
20   .memWrite(MEM_MemWrite),
21   .memRead(MEM_MemRead)
22 );
23 //to WB
24 wire MEM_MemToReg,
25   MEM_RegWrite;
26 assign MEM_MemToReg = EX_MEM_MemToReg;
27 assign MEM_RegWrite = EX_MEM_RegWrite;

```

MEM.v

3.4 阶段间的寄存器更新

每个阶段之间的寄存器在上升沿更新，

```

1 //1.5 IF/ID REG UPDATE
2 always @(posedge CLK) begin
3   IF_ID_PCAdd4 <= IF_PCAdd4;
4   IF_ID_Instr <= IF_Instr;
5 end
6 //2.5 ID/EX REG UPDATE
7 always @(posedge CLK) begin
8   //2.0 For Stage ID To EX
9   ID_EX_PCAdd4 <= ID_PCAdd4;
10  ID_EX_RegReadData1 <= ID_RegReadData1;
11  ID_EX_RegReadData2 <= ID_RegReadData2;
12  ID_EX_SignExt <= ID_SignExt;
13  ID_EX_InstHigh <= ID_InstHigh;
14  ID_EX_InstLow <= ID_InstLow;
15  //2.1 To EX
16  ID_EX_RegDst <= ID_RegDst;
17  ID_EX_ALUOp <= ID_ALUOp;
18  ID_EX_ALUSrc <= ID_ALUSrc;
19  //2.2 To MEM
20  ID_EX_Branch <= ID_Branch;
21  ID_EX_MemWrite <= ID_MemWrite;
22  ID_EX_MemRead <= ID_MemRead;
23  //2.3 To WB
24  ID_EX_MemToReg <= ID_MemToReg;
25  ID_EX_RegWrite <= ID_RegWrite;
26 end
27 //3.5 EX/MEM REG UPDATE

```



```

28 always @(posedge CLK) begin
29     EX_MEM_Branch <= EX_Branch;
30     EX_MEM_MemWrite <= EX_MemWrite;
31     EX_MEM_MemRead <= EX_MemRead;
32     EX_MEM_MemToReg <= EX_MemToReg;
33     EX_MEM_RegWrite <= EX_RegWrite;
34     EX_MEM_BranchAddress <= EX_BranchAddress;
35     EX_MEM_Zero <= EX_Zero;
36     EX_MEM_ALUOut <= EX_ALUOut;
37     EX_MEM_RegWriteAddress <= EX_RegWriteAddress;
38     EX_MEM_RegReadData2 <= EX_RegReadData2;
39 end
40 //4.5 MEM/WB REG UPDATE
41 always @(posedge CLK) begin
42     MEM_WB_ALUOut = MEM_ALUOut;
43     MEM_WB_MemReadData = MEM_MemReadData;
44     MEM_WB_RegWriteAddress = MEM_RegWriteAddress;
45     MEM_WB_RegWrite = MEM_RegWrite;
46     MEM_WB_MemToReg = MEM_MemToReg;
47 end

```

internalRegUpdate.v

4 ModelSim 仿真

1	100011000000000010000000000101100	1	lw \$1, 44(\$0) ; 2
2	1000110000000000100000000000110000	2	lw \$2, 48(\$0) ; 3
3	1000110000000000110000000000110100	3	lw \$3, 52(\$0) ; 4
4	0000000000000000000000000000000000	4	NOP
5	0000000000000000000000000000000000	5	NOP
6	00000000001000100010000000100000	6	add \$4, \$1, \$2 ; \$4=5
7	000000000011000010010100000100010	7	sub \$5, \$3, \$1 ; \$5=2
8	000100000000000000000000000000100	8	beq \$0, \$0, end ;
9	0000000000000000000000000000000000	9	NOP
10	0000000000000000000000000000000000	10	NOP
11	0000000000000000000000000000000000	11	NOP
12	000000000010000110011000000100000	12	add \$6, \$2, \$3 ; not executed
13	10001100000001110000000000101000	13	end: lw \$7, 40(\$0) ; 1
14	00010000000000000000111111111111111	14	beq \$0, \$0, -1 ;
15	0000000000000000000000000000000000	15	NOP
16	0000000000000000000000000000000000	16	NOP

Figure 3: 二进制指令

Figure 4: MIPS 指令




Figure 5: ModelSim 仿真

Part II

改进

实现的最基本的流水线已经能够运行，但在测试程序时很容易发现问题：

1. BEQ 指令后必须插入 NOP，这在 beq 不跳转时会浪费时钟周期；
2. 没有实现 jump 指令；
3. 没有解决 data hazard, 必须插入 NOP 避免 read-after-write；

下面将着手解决这些问题。

5 Flush

5.1 改进方案

执行 beq 指令时，当 IF_Branch 有效之后，下一上升沿将 load 跳转后的指令，此时已经执行到 ID,EX 阶段的指令需要 Flush 掉，因此 ID/EX, EX/MEM, MEM/WB 的所有寄存器清零。下面是修改后的 regUpdate 代码。

```
1 //1.5 IF/ID REG UPDATE
2 always @(posedge CLK) begin
3     IF_ID_PCAAdd4 <= IF_Branch ? 0: IF_PCAAdd4;
4     IF_ID_Instr <= IF_Branch ? 0: IF_Instr;
5 end
6 //2.5 ID/EX REG UPDATE
7 always @(posedge CLK) begin
8     //2.0 For Stage ID To EX
9     ID_EX_PCAAdd4 <= IF_Branch ? 0: ID_PCAAdd4;
10    ID_EX_RegReadData1 <= IF_Branch ? 0: ID_RegReadData1;
11    ID_EX_RegReadData2 <= IF_Branch ? 0: ID_RegReadData2;
12    ID_EX_SignExt <= IF_Branch ? 0: ID_SignExt;
13    ID_EX_InstHigh <= IF_Branch ? 0: ID_InstHigh;
14    ID_EX_InstLow <= IF_Branch ? 0: ID_InstLow;
15    //2.1 To EX
16    ID_EX_RegDst <= IF_Branch ? 0: ID_RegDst;
17    ID_EX_ALUOp <= IF_Branch ? 0: ID_ALUOp;
18    ID_EX_ALUSrc <= IF_Branch ? 0: ID_ALUSrc;
19    //2.2 To MEM
20    ID_EX_Branch <= IF_Branch ? 0: ID_Branch;
21    ID_EX_MemWrite <= IF_Branch ? 0: ID_MemWrite;
22    ID_EX_MemRead <= IF_Branch ? 0: ID_MemRead;
23    //2.3 To WB
24    ID_EX_MemToReg <= IF_Branch ? 0: ID_MemToReg;
25    ID_EX_RegWrite <= IF_Branch ? 0: ID_RegWrite;
26 end
27 //3.5 EX/MEM REG UPDATE
28 always @(posedge CLK) begin
29     EX_MEM_Branch <= IF_Branch ? 0: EX_Branch;
30     EX_MEM_MemWrite <= IF_Branch ? 0: EX_MemWrite;
31     EX_MEM_MemRead <= IF_Branch ? 0: EX_MemRead;
32     EX_MEM_MemToReg <= IF_Branch ? 0: EX_MemToReg;
33     EX_MEM_RegWrite <= IF_Branch ? 0: EX_RegWrite;
34     EX_MEM_BranchAddress <= IF_Branch ? 0: EX_BranchAddress;
35     EX_MEM_Zero <= IF_Branch ? 0: EX_Zero;
36     EX_MEM_ALUOut <= IF_Branch ? 0: EX_ALUOut;
```

```

37     EX_MEM_RegWriteAddress <= IF_Branch ? 0: EX_RegWriteAddress;
38     EX_MEM_RegReadData2 <= IF_Branch ? 0: EX_RegReadData2;
39     end

```

regUpdate.v

5.2 ModelSim 仿真




Figure 6: ModelSim Flush 仿真

1	00000000000000000000000000000000	1	NOP
2	1000110000000001000000000101100	2	lw \$1, 44(\$0) ; 2
3	1000110000000001000000000000110000	3	lw \$2, 48(\$0) ; 3
4	100011000000000110000000000110100	4	lw \$3, 52(\$0) ; 4
5	00000000000000000000000000000000	5	NOP
6	00000000000000000000000000000000	6	NOP
7	00000000000100010001000000010000	7	add \$4, \$1, \$2 ; \$4=5
8	000000000011000010010100000100010	8	sub \$5, \$3, \$1 ; \$5=2
9	00010000000000000000000000000001	9	beq \$0, \$0, end ;
10	00000000010000110011000000100000	10	add \$6, \$2, \$3 ; not executed
11	1000110000000111000000000101000	11	end: lw \$7, 40(\$0) ; 1
12	00010000000000000111111111111111	12	beq \$0, \$0, -1 ;

Figure 7: 二进制指令

Figure 8: MIPS 指令

6 JUMP

6.1 改进方案

实现 jump 需要生成跳转地址，给 nextPC 添加一个 source，为了避免 jump 和 beq 指令的冲突，将 jump 指令延长至 MEM 阶段完成，同时 jump 也需要 Flush 功能。

```

1  \\Def Regs and Wires
2  reg ID_EX_Jump;
3  reg [31:0] ID_EX_Instr;
4  reg EX_MEM_Jump;
5  reg [31:0] EX_MEM_JumpAddress;
6  wire ID_Jump;
7  wire EX_Jump;
8  wire [31:0] EX_JumpAddress;
9  wire IF_Jump;

```

```

10 wire [31:0] IF_JumpAddress;
11 assign EX_Jump = ID_EX_Jump;
12 assign IF_Jump = EX_MEM_Jump;
13 assign EX_Instr = ID_EX_Instr;
14 assign EX_JumpAddress[27:2] = EX_Instr[25:0];
15 assign EX_JumpAddress[1:0] = 2'b00;
16 assign IF_JumpAddress = EX_MEM_JumpAddress;
17 \\Update Regs
18 always @(posedge CLK) begin
19     ID_EX_Jump <= ID_Jump;
20     ID_EX_Instr <= ID_Instr;
21     EX_MEM_Jump <= EX_Jump;
22     EX_MEM_JumpAddress <= EX_JumpAddress;
23 end
24 \\Update PC
25 assign IF_PCAdd4 = IF_CurrPC +4;
26 assign IF_PC Src = IF_Branch & IF_Zero & ~RESET;
27 wire [31:0] IF_NextPCJ;
28 assign IF_NextPCJ = IF_Jump? IF_JumpAddress : IF_PCAdd4;
29 assign IF_NextPC = IF_PC Src? IF_BranchAddress: IF_NextPCJ; //MUX

```

6.2 ModelSim 仿真

1	000010000000000000000000000000001110	1	J 6;
2	00010000000000001111111111111111	2	circle1: beq \$0, \$0, circle1 ;
3	00010000000000000111111111111111	3	circle2: beq \$0, \$0, circle2 ;
4	00010000000000000111111111111111	4	circle3: beq \$0, \$0, circle3 ;
5	00010000000000000111111111111111	5	circle4: beq \$0, \$0, circle4 ;
6	10001100000000010000000000101100	6	lw \$1, 44(\$0) ; 2
7	10001100000000100000000000110000	7	lw \$2, 48(\$0) ; 3
8	10001100000000110000000000110100	8	lw \$3, 52(\$0) ; 4
9	00000000000000000000000000000000	9	NOP
10	00000000000000000000000000000000	10	NOP
11	00000000001000100010000000100000	11	add \$4, \$1, \$2 ; \$4=5
12	00000000011000010010100000100010	12	sub \$5, \$3, \$1 ; \$5=2
13	00010000000000000000000000000001	13	beq \$0, \$0, end ;
14	00000000010000110011000000100000	14	add \$6, \$2, \$3 ; not executed
15	10001100000001110000000001010000	15	end: lw \$7, 40(\$0) ; 1
16	00010000000000001111111111111111	16	beq \$0, \$0, -1 ;

Figure 9: 二进制指令

Figure 10: MIPS 指令




Figure 11: ModelSim JUMP 仿真




Figure 12: top with Forward Unit

7 数据冒险

7.1 MUX

对 ALU 的两个输入增加两个 MUX，其输入分别由信号 ForwardA, ForwardB 控制，控制信号对应的输入如下表所示；将其模块化，代码如下

Mux control	Source	Explanation
ForwardA = 00	ID/EX	The first ALU operand comes from the register file.
ForwardA = 10	EX/MEM	The first ALU operand is forwarded from the prior ALU result.
ForwardA = 01	MEM/WB	The first ALU operand is forwarded from data memory or an earlier ALU result.
ForwardB = 00	ID/EX	The second ALU operand comes from the register file.
ForwardB = 10	EX/MEM	The second ALU operand is forwarded from the prior ALU result.
ForwardB = 01	MEM/WB	The second ALU operand is forwarded from data memory or an earlier ALU result.

Figure 13: MUX 控制信号

```

1 module forwardMux(
2     input [31:0] ID_EX,
3     input [31:0] EX_MEM,
4     input [31:0] MEM_WB,
5     input [1:0] Forward,
6     output [31:0] Sel
7 );
8 wire [31:0] TEMP;
9 assign Sel = Forward[1] ? EX_MEM : TEMP;
10 assign TEMP = Forward[0] ? MEM_WB : ID_EX;
11 endmodule

```

forwardMUX.v

7.2 Forward Unit

对于 R-type 指令引起的 read-after-write, 希望通过如下图所示的方案解决数据冒险。创建模块 ForwardUnit 产生 forwardMux 的控制信号, 实现图示的旁路。




Figure 14: Forwarding
旁路 Forward 逻辑很直观, 见代码内部, 不再单独列出。

```

1 module forwardUnit(
2     input [4:0] rs ,
3     input [4:0] rt ,
4     input MEM_WB_regWrite,
5     input [4:0] MEM_WB_rd,
6     input EX_MEM_regWrite,
7     input [4:0] EX_MEM_rd,
8     output reg [1:0] forwardA ,
9     output reg [1:0] forwardB ,
10    input rst
11);
12 always @(*) begin
13     if(rst | MEM_WB_rd == 0 | EX_MEM_rd == 0) begin
14         forwardA = 'b00;
15         forwardB = 'b00;
16     end
17     else begin
18         if(MEM_WB_regWrite & MEM_WB_rd == rs) forwardA = 'b01;
19         else if(EX_MEM_regWrite & EX_MEM_rd == rs) forwardA = 'b10;
20         else forwardA = 'b00;
21
22         if(MEM_WB_regWrite & MEM_WB_rd == rt) forwardB = 'b01;
23         else if(EX_MEM_regWrite & EX_MEM_rd == rt) forwardB = 'b10;
24         else forwardB = 'b00;
25     end
26 end
27 end
28 endmodule
29

```

forwardUnit.v

7.3 连接信号线及实例化

如顶层的信号连接所示，连接相应的信号。

```
1 //FORWARDING PART
2 //MUX
3 //wire [31:0] EX_ALUSrc1_f; //as mux output;
4 //wire [31:0] EX_ALUSrc2_f; //defined before ALU
5 wire [31:0] EX_MEM_ALU; //input from other stage
6 wire [31:0] MEM_WB_ALU;
7 assign EX_MEM_ALU = EX_MEM_ALUOut;
8 assign MEM_WB_ALU = MEM_WB_ALUOut;
9 wire [1:0] forwardA;
10 wire [1:0] forwardB;
11
12 //FORWARD UNIT
13 wire [4:0] EX_MEM_regWriteAddress_f;
14 wire [4:0] MEM_WB_regWriteAddress_f;
15 wire MEM_WB_regWrite_f;
16 wire EX_MEM_regWrite_f;
17 wire [4:0] rs_f;
18 wire [4:0] rt_f;
19 assign EX_MEM_regWriteAddress_f = EX_MEM_RegWriteAddress;
20 assign MEM_WB_regWriteAddress_f = MEM_WB_RegWriteAddress;
21 assign EX_MEM_regWrite_f = EX_MEM_RegWrite;
22 assign MEM_WB_regWrite_f = MEM_WB_RegWrite;
23 assign rs_f = ID_EX_Instr[25:21]; //ID_EX
24 assign rt_f = ID_EX_Instr[20:16];
25 //instances
26 forwardUnit mainFUnit (
27     .rs(rs_f),
28     .rt(rt_f),
29     .MEM_WB_regWrite(MEM_WB_regWrite_f),
30     .MEM_WB_rd(MEM_WB_regWriteAddress_f),
31     .EX_MEM_regWrite(EX_MEM_regWrite_f),
32     .EX_MEM_rd(EX_MEM_regWriteAddress_f),
33     .forwardA(forwardA),
34     .forwardB(forwardB),
35     .rst(RESET)
36 );
37 forwardMux MUXA (
38     .ID_EX(EX_ALUSrc1),
39     .EX_MEM(EX_MEM_ALU),
40     .MEM_WB(MEM_WB_ALU),
41     .Forward(forwardA),
42     .Sel(EX_ALUSrc1_f)
43 );
44 forwardMux MUXB (
45     .ID_EX(EX_ALUSrc2),
46     .EX_MEM(EX_MEM_ALU),
47     .MEM_WB(MEM_WB_ALU),
48     .Forward(forwardB),
49     .Sel(EX_ALUSrc2_f)
50 );
51 endmodule
```

forwardMUX.v

7.4 ModelSim 仿真

使用 ModelSim 进行仿真，可以看到 data hazard 已经解决。




Figure 15: ModelSim Forwarding Unit

1	00000000001000100000100000100000	1	add \$1, \$1, \$2 ;
2	00000000001000100001000000100000	2	add \$2, \$1, \$2 ;
3	00000000001000100000100000100000	3	add \$1, \$1, \$2 ;
4	00000000001000100001000000100000	4	add \$2, \$1, \$2 ;
5	00000000001000100000100000100000	5	add \$1, \$1, \$2 ;
6	00000000001000100001000000100000	6	add \$2, \$1, \$2 ;
7	7

Figure 16: 二进制指令

Figure 17: MIPS 指令

Part III

上板试验

8 IO Scheme

IO Port	Function
LED[7]	查看 clock 信号
LED[6]	查看 reset 信号
LED[5:0]	查看寄存器数值
SW[3]	调整时钟快/慢模式
SW[2:0]	设置寄存器数值输出模式
Button	输入 reste 信号

Table 1: IO Scheme

SW[2:0]	MODE
000	led[5:3] 为 \$2, led[2:0] 为 \$1
001	led 显示 \$1
011	led 显示 \$2
111	led 显示 PC

设计使用 8 个 LED 作为输出，4 个开关和 1 个复位按钮作为输入，具体功能见下图。因板上接口限制，将只对输出寄存器 \$1,\$2 和 PC, 通过 SW[2:0] 选择输出模式，见左图。

9 IO 控制逻辑

```

1  ///          GENERATING SLOW CLOCK          ///
2  wire CLK;
3  reg [26:0] Buffer = 0;
4  always@ (posedge CLOCK) Buffer = Buffer + 1;
5  assign CLK = FAST ? CLOCK : Buffer[26];
6  ///          IO MODE SEL          ///
7  wire [31:0] IF_CurrPC;
8  assign LED[7] = RESET;
9  assign LED[6] = CLK;
10 wire [5:0] OUTPUT;
11 assign LED[5:0] = OUTPUT;
12 wire [31:0] reg1;
13 wire [31:0] reg2;
14 wire [5:0] reg12;
15 assign reg12[5:3] = reg2;
16 assign reg12[2:0] = reg1;
17 wire [31:0] CURR_PC_IO;
18 assign CURR_PC_IO = IF_CurrPC>>2;
19 wire [5:0] TEMP1;
20 wire [5:0] TEMP2;
21 assign TEMP1 = MODE[0] ? reg1:reg12;
22 assign TEMP2 = MODE[1] ? reg2:TEMP1;
23 assign OUTPUT = MODE[2]? CURR_PC_IO:TEMP2;

```

IO.v

10 User Constraint File

```

1 NET "CLOCK" LOC = C9;
2 NET "LED[0]" LOC = F12;
3 NET "LED[1]" LOC = E12;
4 NET "LED[2]" LOC = E11;
5 NET "LED[3]" LOC = F11;
6 NET "LED[4]" LOC = C11;
7 NET "LED[5]" LOC = D11;
8 NET "LED[6]" LOC = E9;
9 NET "LED[7]" LOC = F9;
10 NET "MODE[0]" LOC = L13;
11 NET "MODE[1]" LOC = L14;
12 NET "MODE[2]" LOC = H18;
13 NET "FAST" LOC = N17;
14 NET "RESET" LOC = K17 | IOSTANDARD = LVTTTL | PULLDOWN;

```

top.ucf